/home/m262f99/CHYZAK/worksheetsV.4/hw8ansm.mws

Math 262a, Fall 1999, Glenn Tesler
Homework 8
11/24/99

Problem 3. We have two functions defined by recurrence equations:

```
> n*f(n+1)-f(n)=0;
  g(n+2)-n*g(n)=0;
```

$$n\,\mathrm{f}(n+1)-\mathrm{f}(n)=0$$

$$\mathrm{g}(n+2)-n\,\mathrm{g}(n)=0$$

We want to find a recurrence equation satisfied by

```
> h(n) := f(n)*g(n)-g(n);
```

$$\mathrm{h}(n) := \mathrm{f}(n)\,\mathrm{g}(n)-\mathrm{g}(n)$$

Build a table of shifts of h(x), reduced by the recurrence equations that f(n),g(n) satisfy. The result will only involve f(n), g(n), g(n+1), and no other shifts:

```
> myreduce := proc(fn)
     subs(f(n+1)=-f(n)/n,
  g(n+2)=n*g(n),g(n+3)=(n+1)*g(n+1), fn)
  end:
> dh[0] := h(n);
  for k from 1 to 6 do
        dh[k] := myreduce(subs(n=n+1,dh[k-1]));
  od; k := 'k': # reset k to unevaluated symbol
```

$$dh_0 := \mathrm{f}(n)\,\mathrm{g}(n)-\mathrm{g}(n)$$

$$dh_1 := -\frac{\mathrm{f}(n)\,\mathrm{g}(n+1)}{n}-\mathrm{g}(n+1)$$

$$dh_2 := \frac{\mathrm{f}(n)\,\mathrm{g}(n)}{n+1}-n\,\mathrm{g}(n)$$

$$dh_3 := -\frac{\mathrm{f}(n)\,\mathrm{g}(n+1)}{n\,(n+2)}-(n+1)\,\mathrm{g}(n+1)$$

$$dh_4 := \frac{\mathrm{f}(n)\,\mathrm{g}(n)}{(n+1)\,(n+3)}-(n+2)\,n\,\mathrm{g}(n)$$

$$dh_5 := -\frac{\mathrm{f}(n)\,\mathrm{g}(n+1)}{n\,(n+2)\,(n+4)}-(n+3)\,(n+1)\,\mathrm{g}(n+1)$$

$$dh_6 := \frac{f(n)\, g(n)}{(n+1)\,(n+3)\,(n+5)} - (n+4)\,(n+2)\, n\, g(n)$$

Symbollically, all of the above are in the span of the following elements:

```
> B_module := [ f(n)*g(n), f(n)*g(n+1), g(n), g(n+1) ];
```

$$B\_module := [\, f(n)\, g(n),\, f(n)\, g(n+1),\, g(n),\, g(n+1)\,]$$

(Note: it's possible that when we plug in the solutions for f(n), g(n), that the above would be dependent, but that's irrelevant. We treat these as formal, linearly independent symbols, and the true values of these give a space that's a "homomorphic image" of the symbollic space we work in.)

which implies that any 5 of these are linearly dependent (5>4), so we take dh[0]..dh[4].

Form a dependency equation.

```
> dependency_symbollic := a[0]*‘ h‘(n) + sum(a[k]*‘
  h‘(n+k),k=1..4);
  dependency := sum(a[k]*dh[k],k=0..4);
```

dependency_symbollic :=

$$a_0\ h(n) + a_1\ h(n+1) + a_2\ h(n+2) + a_3\ h(n+3) + a_4\ h(n+4)$$

$$dependency := a_0\,(f(n)\, g(n) - g(n)) + a_1 \left(-\frac{f(n)\, g(n+1)}{n} - g(n+1) \right) + a_2 \left(\frac{f(n)\, g(n)}{n+1} - n\, g(n) \right) + a_3 \left(-\frac{f(n)\, g(n+1)}{n\,(n+2)} - (n+1)\, g(n+1) \right) + a_4 \left(\frac{f(n)\, g(n)}{(n+1)\,(n+3)} - (n+2)\, n\, g(n) \right)$$

\>

The maple command collect is for polynomials, not modules, so it doesn't like the form of our basis B:

```
> collect(dependency,B_module);
Error, (in collect) cannot collect, f(n)*g(n)
```

We could replace everything in B by symbols f(n)*g(n)->b1, f(n)*g(n+1)->b2, etc., resulting in linear polynomials in the b's, which are the same as a module spanned by the b's. But instead, since the basis is generated by polynomials in f(n), g(n), g(n+1) we do:

```
> B_2 := [f(n),g(n),g(n+1)];
  collect(dependency,B_2,distributed);
```

$$B\_2 := [\, f(n),\, g(n),\, g(n+1)\,]$$

$$\left(a_0 + \frac{a_2}{n+1} + \frac{a_4}{(n+1)\,(n+3)} \right) g(n)\, f(n) + \left(-\frac{a_1}{n} - \frac{a_3}{n\,(n+2)} \right) g(n+1)\, f(n) + (-a_0 - a_2\, n - a_4\,(n+2)\, n)\, g(n) + (-a_1 - a_3\,(n+1))\, g(n+1)$$

```
> B_cofs := coeffs(",B_2);
```

$$B\_cofs := a_0 + \frac{a_2}{n+1} + \frac{a_4}{(n+1)(n+3)}, -a_1 - a_3(n+1), -a_0 - a_2 n - a_4(n+2)n,$$
$$-\frac{a_1}{n} - \frac{a_3}{n(n+2)}$$

```
> B_sol := solve({B_cofs},{'a[k]'$k=0..4});
```

$$B\_sol := \{a_3 = 0, a_2 = a_2, a_1 = 0, a_4 = -\frac{a_2(n^3+4n^2+2n-3)}{-1+n^4+6n^3+11n^2+6n},$$
$$a_0 = -\frac{a_2 n(5+n^2+5n)}{-1+n^4+6n^3+11n^2+6n}\}$$

So the final recurrence equation is

```
> hDE := subs(B_sol,dependency_symbollic):
  hDE = 0;
```

$$-\frac{a_2 n(5+n^2+5n)\ \mathrm{h}(n)}{-1+n^4+6n^3+11n^2+6n} + a_2\ \mathrm{h}(n+2) - \frac{a_2(n^3+4n^2+2n-3)\ \mathrm{h}(n+4)}{-1+n^4+6n^3+11n^2+6n} = 0$$

where a[2] is anything at all. Let's clear denominators:

```
> hDE := subs(a[2]=denom(hDE),hDE):
  map(factor,hDE) = 0;
```

$$-n(5+n^2+5n)\ \mathrm{h}(n) + (-1+n^4+6n^3+11n^2+6n)\ \mathrm{h}(n+2)$$
$$-(n+3)(n^2+n-1)\ \mathrm{h}(n+4) = 0$$

Chyzak's routine to compute the same end result as the above, but using Grobner bases.:

First reset all variable names.

```
> restart;
```

```
> with(Groebner): with(Ore_algebra): with(Holonomy):
  with(Mgfun):
```

```
> pol_to_sys(h(n)=f(n)*g(n)-g(n),
       {[f(n), {n*f(n+1)-f(n)}], [g(n),
  {g(n+2)-n*g(n)}]});
```

$$\{(n^3+4n^2+2n-3)\,\mathrm{h}(n+4) + (5n^2+5n+n^3)\,\mathrm{h}(n)$$
$$+(-6n^3-n^4-6n-11n^2+1)\,\mathrm{h}(n+2)\}$$

```
> map(factor,op("));
```

$$(n+3)(n^2+n-1)\,\mathrm{h}(n+4) + n(5n+5+n^2)\,\mathrm{h}(n)$$
$$-(n^4+6n^3+11n^2+6n-1)\,\mathrm{h}(n+2)$$

(The =0 is implied; this is the negative of the recurrence we had the other way.)

[>